File Searching With the Microsoft Index Service

Takeaway: Search is a key aspect of any Web site design that all users expect. The Microsoft Index Service makes it easy to implement and utilize a search feature on your Web site.
The Microsoft Index Service provides a powerful full text indexing service for either the entire file system or just Web files. Querying the Index Service is straightforward, and the Index Service provides a foundation for delivering a more robust Web site. You can query the Index Service with the Index Service Query Language or with SQL queries, both of which return ADO recordsets.

The built-in Query object allows you to easily add impressive search features to our Web sites. Let's take a look at how to use the Index Service Query Language and Index Service Query object.

Index Service's Query object
The Index Service’s Query object provides an interface for querying the index’s catalog and returning an ADO recordset. The Query object contains several properties to control aspects of the query and how the data is returned. First, you must instantiate the query object in an ASP page like this:
<%
set objQuery = Server.CreateObject(“ixsso.Query”)
%>
Query object properties
There are several properties of this object that you can use to refine your query. In the following sections, I will discuss each property and explain how to set them.

Catalog
The Catalog property defines which index catalog is used in the query. The Index Service can be configured to have multiple catalogs on a single server. For example, a server may have a catalog of the Web data that is only in the INETPUB directory and another catalog of the entire system. The Query object defaults to the system catalog unless IIS is installed on the computer. If IIS is installed, the default is the Web catalog.

In addition to querying local catalogs, you can also query remote servers’ catalogs. The syntax for querying a remote catalog is query://hostname/catalog. The following are a few examples of setting the Catalog property:
‘Query the Web catalog on the local system
objQuery.Catalog = “web”
‘Query the System catalog on the local system
objQuery.Catalog = “system”
 ‘Query the Web catalog on the computer www.somehost.com
objQuery.Catalog = “query://www.somehost.com/web”
Columns
With any query, you want to control the data that is returned. The Columns property allows you to define the fields that are returned for each search result. Table A contains a list of the columns available and what information the columns contain.

Table A
	Column 

	Contents 


	Characterization

	Abstract of the file created by the Indexing Service


	DocTitle

	The document title—for HTML documents, it is the text contained in the <TITLE> tag


	Filename

	The name of the file


	Path

	Full path of the file including the file name


	Rank

	Numeric value of how closely the file matches the query


	Size

	Size of the file in bytes


	Vpath

	Virtual path of the file including the file name


	Write

	The last time the file was written



	



The Columns property is set in this way:
objQuery.Columns = “Characterization, DocTitle, Filename, Path, Rank, Size, Vpath, Write”
SortBy
The SortBy property defines how the recordset is to be sorted. The SortBy property can be set to any of the columns that have been selected for the recordset. Also, you can define whether you want the results to be sorted in ascending or descending order by appending either [a] or [d] to the column name. The default sort order is ascending. The follow example sorts the results by rank in descending order and then by size in ascending order:
objQuery.SortBy = “Rank[d], size”
Defining where to search and how deeply
Even though you can define the catalog you want to query, sometimes you want to query only one directory in the entire catalog. You can do this by defining a query scope and depth. To define the query scope, call the AddScopeToQuery method in the Index Service Util object. The AddScopeToQuery method accepts three parameters: first, the query object to modify; second the scope to be used; and finally, the depth of the search. The scope is the name of the directory from which the search is to begin. For example, if I wanted to search the Samples directory, I would set the scope to /samples. The depth of the search accepts two values, shallow or deep. A shallow query queries only the directory named as the scope, while a deep query searches the named directory and all of its subdirectories. The following example demonstrates how to instantiate the Util object and how to call the AddScopeToQuery method:
Set objUtil = Server.CreateObject(“ixsso.Util”)
ObjUtil.AddScopeToQuery objQuery, “/samples”, “deep”
The query
Once you've defined where to search, how much to search, and how to order the results, you can write the actual query. For these examples, I used the Index Service Query Language Short Form, a compact, easy-to-use query language. Let’s look at a few keywords from this language.

Contains
The Contains keyword defines words or phrases that you want to find in the body of the target documents. Table B presents ways to use the Contains keyword.
Table B
	@contains web server software

	Search for documents containing the words web, server, and software 


	@contains “web server software”

	Search for documents containing the phrase “web server software” 


	@contains web near software

	Search for documents that have the word web near to the word software. The resulting documents are ranked by how close the words are to each other



	



Write
If you want to find files that have been modified since a certain time, you may use the Write keyword. The unique aspect of the Write keyword is the ability to utilize a time parameter in the search. Table C shows how to format the query.
Table C
	Search Term 

	Results 


	@write > -1n

	Documents modified in the last minute


	@write > -1h

	Documents modified in the last hour


	@write > -1d

	Documents modified in the last day


	@write > -1w

	Documents modified in the last week


	@write > -1m

	Documents modified in the last month


	@write > -1q

	Documents modified in the last quarter


	@write > -1y

	Documents modified in the last year


	@write 02/01/11

	Documents modified since January 11, 2002



	



Size
The Size keyword allows you to search for documents of a particular size. Take a look at the example in Table D.
Table D

	@size > 1024

	Returns all documents larger than one kilobyte


	@size > 1024 & < 1048576

	Returns all documents larger than one kilobyte but smaller than one megabyte



	



Now, let’s put together a search string with the Query object:
objQuery.Query = “(@contents “”web server software””) & (@write > -1q)”
This query will return the documents that contain the phrase “web server software” and that have been modified in the last quarter.

Put it all together
Now that you've defined what you want to find, you can create a record set with the CreateRecordset method of the Query object like this:
Set RecSet = objQuery.CreateRecordset(“nonsequential”)
The parameter for CreateRecordset is either sequential or nonsequential. The resulting recordset can then be used to retrieve the results of the query. The following example demonstrates an entire ASP page for a simple query:
<%
set objQuery =Server.CreateObject("ixsso.Query")
objQuery.Catalog = "web"
objQuery.Columns = "Characterization, DocTitle, Filename, Path, Rank, Size, Vpath, Write"
objQuery.SortBy = "Rank[d], size"

Set objUtil = Server.CreateObject("ixsso.Util")
objUtil.AddScopeToQuery objQuery, "/", "deep"
objQuery.Query = "(@contents book) & (@write > -1y)"

set RecSet = objQuery.CreateRecordset("nonsequential")

while not RecSet.EOF
Response.Write "<A HREF=""" & RecSet("vPath") & """>" & RecSet("DocTitle") & "</A><BR>" & vbCrLf
Response.Write "<b>Summary: </b>" & RecSet("Characterization") & "<BR>" & vbCrLf
Response.Write "<b>Rank: </b>" & RecSet("Rank") & "<BR>" & vbCrLf
Response.Write "<b>Size: </b>" & RecSet("Size") & "<BR>" & vbCrLf
Response.Write "<b>Last Update: </b>" & RecSet("Write") &  "<HR>" & vbCrLf
RecSet.MoveNext
wend
%>
This ASP page produces a simple list of the results of the query and includes a link to the document, a summary of the document, the rank of how closely the document matches the query, the size of the file, and the date of when the file was last updated.

Power at your fingertips
The Microsoft Index Service provides powerful indexing and search capabilities right out of the box with Windows 2000. Its built-in Query object lets you more easily add impressive search features to your Web sites. The tips I've provided should help you get started using this handy feature.

